Pensieve header: A unified verification notebook for the PPSA project.

Continues pensieve://2017-06/ and pensieve://2017-08/.

Prolog

```
Go;
```

```
SetOptions[EvaluationNotebook[],
  NotebookEventActions → {
    {"MenuCommand", "Save"} :→ (
      SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\PPSA"];
      Get["MakeVSnips.m"]
     ),
    PassEventsDown → True
   }];
```

```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Initialization

The “degree carrier / filtration parameter” is $\hbar$, and all “coupling constants” are proportional to it.

TD

```
$TℏD = 5; ℏ /: ℏ^d_. /; d > $TℏD := 0;
$TϵD = 3; ϵ /: ϵ^d_. /; d > $TϵD := 0;
SetAttributes[SS, HoldAll];
SS[ℰ_] := Block[{ℏ, ϵ},
  Normal@Series[ℰ, {ℏ, 0, $TℏD}, {ϵ, 0, $TϵD}] ]
Simp[ℰ_] := Collect[ℰ, _CU | _QU, Expand];
```

DeclareAlgebra

QLImplementation

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x; 0 ** _ = _ ** 0 = 0;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

QLImplementation

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, cp, S, pow,
    gs = Generators /. {opts}, cs = Centrals /. {opts}
  },
  gp = Alternatives @@ gs;
  gp = gp | gp_; (* generators pattern *)
  sr = Thread[gs → Range@Length@gs]; (* sorting rule *)
  cp = Alternatives @@ cs; (* centrals pattern *)
  S[ℰ_] := Collect[ℰ, _U, Expand];
  U_i_[ℰ_] := ℰ /. {t : cp ⧴ t_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
  B[U@(x_)_i_, U@(y_)_i_] := B[U@x_i, U@y_i] = U_i@B[U@x, U@y];
  B[U@(x_)_i_, U@(y_)_j_] /; i =!= j := 0;
  B[U@y_, U@x_] := S[-B[U@x, U@y]];
  x_ ** U[] := x; U[] ** x_ := x;
  (a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, S[a b (x ** y)]];
  (a_ * x_U) ** y_ := S[a (x ** y)]; x_ ** (a_ * y_U) := S[a (x ** y)];
  U[xx___, x_] ** U[y_, yy___] := If[OrderedQ[{x, y} /. sr],
    U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
  U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := S[c U@Table[l, {n}] ** U@{r}];
  U@{c_. * l : gp, r___} := S[c U[l] ** U@{r}];
  U@{c_, r___} /; FreeQ[c, gp] := S[c U@{r}];
  U@{} = U[];
  U@{l_Plus, r___} := S[U@{#, r} & /@ l];
  U@{l_, r___} := U@{Expand[l], r};
  𝕆_U[poly_, specs___] := Module[{sp, null, vs, us},
    sp = Replace[{specs}, l_List ⧴ (l → null), {1}];
    vs = Join @@ (First /@ sp);
    us = Join @@ (sp /. (l_ → s_) ⧴ (l /. x_i_ ⧴ x_s));
    S[Total[
       CoefficientRules[poly, vs] /. (p_ → c_) ⧴ c U@(us^p)
     ]] /. x_null ⧴ x
   ];
  pow[ℰ_, 0] = U[]; pow[ℰ_, n_] := pow[ℰ, n - 1] ** ℰ;
  𝕊_U[ℰ_, ss___Rule] := S@Total[
    CoefficientRules[ℰ, First /@ {ss}] /.
     (p_ → c_) ⧴ c NonCommutativeMultiply @@ MapThread[pow, {Last /@ {ss}, p}]]
  ]
```

DeclareMorphism

QLImplementation

```
DeclareMorphism[m_, U_ → V_, ongs_List, oncs_List: {}] := (
  Replace[ongs, (g_ → img_) ⧴ (m[U[g]] = img), {1}];
  m[U[]] = V[];
  m[U[g_i_]] := V_i[m[U@g]];
  m[U[vs__]] := NonCommutativeMultiply @@ (m /@ U /@ {vs});
  m[ℰ_] := Simp[ℰ /. oncs /. u_U ⧴ m[u]]; )
```

Implementing $sl_2^{\gamma\epsilon}$

CU

```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[CU@a, CU@y] = -γ CU@y; B[CU@x, CU@a] = -γ CU@x;
B[CU@x, CU@y] = 2 ϵ CU@a - t CU[];
```

Verifying associativity on triples of generators:

```
With[{bas = CU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple:

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp
  }] // Timing
```

```
{0.625,
 {(28 t^2 γ^4 + 116 t γ^5 ϵ + 120 γ^6 ϵ^2) CU[y, y, y, x, x] + <<23>> + CU[y, y, y, y, <<6>>, x, x, x], 0}}
```

Implementing $\mathcal{U}_{\gamma\epsilon;\hbar}$

With $q = e^{\hbar\gamma\epsilon}$, $A = e^{-\hbar\epsilon a}$, and $[f, g]_q := fg - qgf$, our algebra is $\mathcal{U}_{\gamma\epsilon;\hbar} = \langle t, y, a, x\rangle / \mathcal{R}$, where $\mathcal{R} = \left([t, *] = 0,\ [a, y] = -\gamma y,\ [x, y]_q = \left(1 - T^2 A^2\right)/\hbar,\ [x, a] = -\gamma x\right)$.

QU

```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
q := Normal@Series[e^(γ ϵ ħ), {ħ, 0, $TħD}];
B[QU@a, QU@y] = -γ QU@y; B[QU@x, QU@a] = -γ QU@x;
B[QU@x, QU@y] := (q - 1) QU@{y, x} + 𝕆_QU[SS[(1 - T^2 e^(-2 ħ ϵ a)) / ħ], {a}];
```

```
With[{bas = QU /@ {y, a, x}}, Table[z1 ** z2 - z2 ** z1 // Simp, {z1, bas}, {z2, bas}] ]
```

$$\left\{\left\{0,\ \gamma\,QU[y],\ \left(-\frac{1}{\hbar}+\frac{T^2}{\hbar}\right)QU[\,] - 2T^2\epsilon\,QU[a] + 2T^2\epsilon^2\hbar\,QU[a,a] + \left(-\gamma\epsilon\hbar - \frac{1}{2}\gamma^2\epsilon^2\hbar^2\right)QU[y,x]\right\},\right.$$
$$\{-\gamma\,QU[y],\ 0,\ \gamma\,QU[x]\},$$
$$\left.\left\{\left(\frac{1}{\hbar}-\frac{T^2}{\hbar}\right)QU[\,] + 2T^2\epsilon\,QU[a] - 2T^2\epsilon^2\hbar\,QU[a,a] + \left(\gamma\epsilon\hbar + \frac{1}{2}\gamma^2\epsilon^2\hbar^2\right)QU[y,x],\ -\gamma\,QU[x],\ 0\right\}\right\}$$

Verifying associativity on triples of generators:

```
With[{bas = QU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple (~34 secs):

```
With[{z1 = QU[y, y, a, a, x, x], z2 = QU[y, a, x], z3 = QU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp
  }] // Timing
```

$$\{49.5938,\ \left\{\left(135\,\gamma^6\,\epsilon^2 - 730\,T^2\,\gamma^6\,\epsilon^2 + \ll 8 \gg + \frac{198\,T^4\,\gamma^5\,\epsilon}{\hbar}\right) QU[y, y, y, x, x] + \ll 22 \gg + (\ll 1 \gg)\ \ll 1 \gg,\ 0\right\}\}$$

Verifying that $\mathrm{Lim}_{\hbar\to 0}$ QU = CU using a “random” product (~23 secs):

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   Short[lhs = z1 ** (z2 ** z3)],
   Short[rhs = (QU @@ z1) ** ((QU @@ z2) ** (QU @@ z3))],
   Expand[Limit[rhs /. {QU → CU, T → ⅇ^(ℏ t/2)}, ℏ → 0] == lhs]
  }] // Timing
```

$$\{28.0156,\ \{2\,(8\,t^2\,\gamma^4 + 16\,t\,\gamma^5\,\epsilon)\ CU[y, y, y, x, x] + \ll 94 \gg + CU[y, y, y, y, y, a, a, a, a, x, x, x, x],$$
$$(-8\,T^2\,\gamma^6\,\epsilon^2 + 8\,T^4\,\gamma^6\,\epsilon^2)\ QU[y, y, y, x, x] + \ll 422 \gg + \left(\gamma\,\epsilon\,\hbar + \frac{15}{2}\,\gamma^2 \ll 1 \gg \hbar^2\right) \ll 1 \gg,\ True\}\}$$

Implementing θ

theta

```
DeclareMorphism[Cθ, CU → CU, {y → -CU@x, a → -CU@a, x → -CU@y}, {t → -t, T → T^-1}];
DeclareMorphism[Qθ, QU → QU, {y → 𝕆_QU[SS[-T^-1 ⅇ^(ℏ ϵ a) x], {a, x}],
  a → -QU@a, x → 𝕆_QU[SS[-T^-1 ⅇ^(ℏ ϵ a) y], {a, y}]}, {t → -t, T → T^-1}]
```

Verifying involutivity on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[z → Cθ[z] → Cθ[Cθ[z]], {z, bas}]]
```

```
{CU[y] → -CU[x] → CU[y], CU[a] → -CU[a] → CU[a], CU[x] → -CU[y] → CU[x]}
```

Verifying that θ is a multiplicative homomorphism on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cθ[z1 ** z2] - Cθ[z1] ** Cθ[z2], {z1, bas}, {z2, bas}]]
```

```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Verifying involutivity on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[z → Qθ[z] → HL[Qθ[Qθ[z]]], {z, bas}] ]
```

$$\left\{QU[y] \to -\frac{QU[x]}{T} - \frac{\epsilon\,\hbar\,QU[a,x]}{T} - \frac{\epsilon^2\,\hbar^2\,QU[a,a,x]}{2\,T} \to QU[y],\ QU[a] \to -QU[a] \to QU[a],\right.$$
$$\left. QU[x] \to \left(-\frac{1}{T} + \frac{\gamma\,\epsilon\,\hbar}{T} - \frac{\gamma^2\,\epsilon^2\,\hbar^2}{2\,T}\right) QU[y] + \left(-\frac{\epsilon\,\hbar}{T} + \frac{\gamma\,\epsilon^2\,\hbar^2}{T}\right) QU[y,a] - \frac{\epsilon^2\,\hbar^2\,QU[y,a,a]}{2\,T} \to QU[x]\right\}$$

Verifying that θ is a multiplicative homomorphism on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → Simp[Qθ[z1 ** z2] - Qθ[z1] ** Qθ[z2]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Asymmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

ADeq

$$\text{ADeq\$f} = e^{\hbar\,(t/2-\epsilon-\epsilon\,a)} \left(\left(\text{Cosh}\left[\hbar\left(\frac{\epsilon-t}{2}+\epsilon\,a\right)\right] - \text{Cosh}\left[\hbar\sqrt{\left(\frac{t-\epsilon}{2}\right)^2+\epsilon\,\omega}\,\right]\right) \Big/ \left(\hbar\,\text{Sinh}\left[\frac{-\epsilon\,\hbar}{2}\right]\left(\omega-\epsilon\,a^2+(t-\epsilon)\,a\right)\right)\right) y;$$

ADeq

$$\text{DeclareMorphism}\Big[\text{ADeq},\ QU \to CU,\ \Big\{a \to CU@a,\ x \to CU@x,\ y \to \mathbb{S}_{CU}\Big[SS[\text{ADeq\$f} \mathbin{/.} \{t \to \delta_7\,t,\ \epsilon \to \delta_8\,\epsilon\}],\ a \to \delta_1\,CU[a],\ \omega \to \delta_2\,CU[y,x] + \delta_3\,\epsilon\,CU[a,a] - (\delta_4\,t - \delta_5\,\epsilon)\,CU[a],\ y \to \delta_6\,CU[y]\Big]\Big\}\Big]$$

```
{Short[tt = ADeq[QU[y, y, a, a, x, x]]], tt /. ħ → 0}
```

```
{≪1≫, CU[y, y, a, a, x, x]}
```

$$\text{With}\Big[\{bas = QU \mathbin{/@} \{y, a, x\}\},\ \text{Table}\Big[\{z1, z2\} \to \text{Simp}\Big[\text{ADeq}[z1 ** z2] - \text{ADeq}[z1] ** \text{ADeq}[z2] \mathbin{/.} \Big\{\gamma \to 1,\ T \to SS\big[e^{\hbar\,t/2}\big]\Big\}\Big],\ \{z1, bas\},\ \{z2, bas\}\Big]\Big]$$

$$\Big\{\{\{QU[y], QU[y]\} \to 0,\ \{QU[y], QU[a]\} \to 0,\ \{QU[y], QU[x]\} \to 0\},$$
$$\{\{QU[a], QU[y]\} \to 0,\ \{QU[a], QU[a]\} \to 0,\ \{QU[a], QU[x]\} \to 0\},$$
$$\Big\{\{QU[x], QU[y]\} \to \frac{t^6\,\hbar^5\,CU[\,]}{23\,040},\ \{QU[x], QU[a]\} \to 0,\ \{QU[x], QU[x]\} \to 0\Big\}\Big\}$$

```
With[{bas = QU /@ {y, a, x}},
  Table[{z1, z2} → Simp[
     ADeq[z1 ** z2] - ADeq[z1] ** ADeq[z2] /. {T → SS[ⅇ^(ℏ t/2)]}
    ], {z1, bas}, {z2, bas}] ] /.
 {δ6 → 1, δ7 → 1, δ8 → δ1^-1, δ4 → γ^-1, δ2 → 1, δ5 → γ δ3} /. {δ1 → γ^-1, δ3 → γ^-1}
```

$$\left\{\{\{QU[y], QU[y]\} \to 0, \{QU[y], QU[a]\} \to 0, \{QU[y], QU[x]\} \to 0\}, \{\{QU[a], QU[y]\} \to 0, \{QU[a], QU[a]\} \to 0, \{QU[a], QU[x]\} \to 0\}, \left\{\{QU[x], QU[y]\} \to \frac{t^6 \hbar^5 CU[]}{23040}, \{QU[x], QU[a]\} \to 0, \{QU[x], QU[x]\} \to 0\right\}\right\}$$